
Goal-Directed Proof

Tacticals

Tacticals

x Tacticals are ML functions used to:
u build more powerful tactics from simpler ones,

u modify the behaviour of tactics,

u construct complete goal-directed proofs.

x For example, sequencing of tactics:

x We will now look at the basic HOL tacticals. . .

Choice of Alternatives

x The tactical ORELSE is an infix ML function:

ORELSE : tactic -> tactic -> tactic

x If T1 and T2 are tactics, then

T1 ORELSE T2

is a tactic which

u first tries the tactic T1, and

u applies T1 if it succeeds, or else

u tries T2 if T1 fails

x The ML definition of ORELSE is simple:

- infix ORELSE;
infix ORELSE

- fun (f1 ORELSE f2) g = f1 g handle _ => f2 g;
> val ORELSE = fn:(’a -> ’b) * (’a -> ’b) -> ’a -> ’b

Example of Choice

x Example, try CONJ_TAC or else GEN_TAC

- val tac:tactic = CONJ_TAC ORELSE GEN_TAC;
> val tac = fn : tactic

- set_goal ([], ‘‘!x:’a. R x‘‘);
> val it = New goal stack.
!x. R x
____________________________

: proofs

- expand CONJ_TAC;
OK..
uncaught exception HOL_ERR

- expand tac;
OK..
1 subgoal:
> val it =
R x
____________________________

: proofs

Sequencing

x The tactical THEN is an ML infix:

THEN : tactic -> tactic -> tactic

x If T1 and T2 are tactics, then

T1 THEN T2

is a tactic which

u first applies the tactic T1, and

u then applies T2 to all the resulting subgoals

x In a picture, T1 THEN T2:

g

T1

([g1,...,gn], f)

6T2 6T2

([g11,...,g1i,...,gn1,...,gnj], F)

Example of Sequencing

x Example:

- set_goal([], ‘‘(!x:num. R x) /\ (!y:num. Q y)‘‘);
> val it = New goal stack.
(!x. R x) /\ (!y. Q y)
____________________________

: proofs
- expand CONJ_TAC;
OK.. 2 subgoals:
> val it =
!y. Q y
____________________________

!x. R x
____________________________

: proofs
- backup();
> val it = New goal stack.
(!x. R x) /\ (!y. Q y)
____________________________

: proofs
- expand (CONJ_TAC THEN GEN_TAC);
OK.. 2 subgoals:
> val it =
Q y
____________________________

R x
____________________________

: proofs

Alternative Sequencing

x THEN applies its second tactic to all the
subgoals generated by the first tactic.

x If you want to do something different to each
resulting subgoals, use THENL:

THENL : tactic -> tactic list -> tactic

x If T , T1, . . . , Tn are tactics, then

T THENL [T1,. . .,Tn]

is a tactic which

u first applies the tactic T , and

u then applies Ti to the ith resulting subgoal

x If T generates n subgoals, then there must be
exactly n tactics in the supplied list.

Repetition

x Tactical that repeats a tactic until failure:

REPEAT : tactic -> tactic

x If T is a tactic, then the tactic REPEAT T

applies T repeatedly until it fails.

x The ML definition of REPEAT:

- val ALL_TAC : tactic = fn g => ([g],hd);
> val ALL_TAC = fn : tactic

- fun REPEAT tac g =
((tac THEN REPEAT tac) ORELSE ALL_TAC) g ;

> val REPEAT = fn : tactic -> tactic

Example of Repetition

x A very common first step:

- set_goal ([], ‘‘!R Q. ~R \/ Q ==> (R ==> Q)‘‘);
> val it = New goal stack.
!R Q. ~R \/ Q ==> R ==> Q
____________________________

: proofs

- expand (REPEAT STRIP_TAC);
OK.. 2 subgoals:
> val it =
Q
____________________________

Q
R

Q
____________________________

~R
R : proofs

Tacticals and the Subgoal Package

x There are two distinctly different things in goal-
directed proof:

u the activity of finding a proof

u the completed proof itself

x The subgoal package, and completed proofs:
'

&

$

%
HOL session ML source

expand ;

expand ;

expand ;
...

THEN

THEN
...

x HOL provides some auxiliary functions for
packaging-up completed tactic proofs. . .

Proving and Saving a Theorem

x To prove a theorem and save it:

store_thm : (string * term * tactic) -> thm

This function:

u takes as arguments name ‘〈name〉‘, a goal (A, g)

and a tactic T ,

u tries to prove the goal using the tactic,

u if the tactic succeeds in proving the goal, returns

the resulting theorem, and

u saves the theorem under the name ‘〈name〉‘ in the

current theory.

Example of Packaged Proofs

x Part of the ML sources for the built-in HOL
theory of arithmetic:

...

val ADD_SYM =
store_thm
("ADD_SYM",
‘‘!m n. m + n = n + m‘‘,
Induct THEN
GEN_TAC THEN
ASM_REWRITE_TAC[ADD_0,ADD,ADD_SUC]);

val num_CASES =
store_thm
("num_CASES",
‘‘!m. (m = 0) \/ ?n. m = SUC n‘‘,
Induct THEN
REWRITE_TAC[NOT_SUC] THEN
EXISTS_TAC (‘‘m:num‘‘) THEN
REWRITE_TAC[]);

...

Summary of Tacticals

x Choice of alternatives:

ORELSE : tactic -> tactic -> tactic

x Sequencing:

THEN : tactic -> tactic -> tactic

x Alternative sequencing:

THENL : tactic -> tactic list -> tactic

x Repetition:

REPEAT : tactic -> tactic

x Auxiliary functions:

TAC_PROOF : (goal * tactic) -> thm

store_thm : (string * goal * tactic) -> thm

List Tacticals

x Two simple list tacticals are:

EVERY : tactic list -> tactic

FIRST : tactic list -> tactic

x They behave as follows:

u EVERY [T1,...,Tn] applies the tactics T1, . . . , Tn in

sequence to the goal.

u FIRST [T1,...,Tn] applies the first of the tactics

T1, . . . , Tn that succeeds when applied to the goal.

x Thus we have the equivalences:

EVERY [T1,...,Tn] = T1 THEN ... THEN Tn

FIRST [T1,...,Tn] = T1 ORELSE ... ORELSE Tn

Mapping Tacticals

x The tacticals

MAP_EVERY : (’a -> tactic) -> ’a list -> tactic

MAP_FIRST : (’a -> tactic) -> ’a list -> tactic

take an ML function f of type ’a -> tactic and
a list l of type ’a list and

u apply f all the elements in the list l to get a list

of tactics, and

u apply every tactic in the list (resp. the first tactic

that succeeds) to the goal.

x For example

MAP_EVERY ASSUME_TAC [|- P1, ..., |-Pn]

is equivalent to

EVERY [ASSUME_TAC |- P1, ..., ASSUME_TAC |- Pn]

Using the Assumptions

x There are three list tacticals for using the
assumption of a goal:

EVERY_ASSUM : (thm -> tactic) -> tactic

FIRST_ASSUM : (thm -> tactic) -> tactic

ASSUM_LIST : (thm list -> tactic) -> tactic

x These use the assumptions of a goal as theorems.

x For example, use the assumptions as inputs to
tactics like ACCEPT_TAC or CONTR_TAC.

Using Every Assumption

x The tactical

EVERY_ASSUM : (thm -> tactic) -> tactic

takes an ML function f of type thm -> tactic

and a goal (A,g) and

u maps ASSUME down the list of assumptions A,

u maps f down the resulting list of theorems,

u applies every tactic in the resulting list of tactics

to the goal.

x You can read

EVERY_ASSUM ttac

as ‘do ttac, using every assumption’.

Example

x Consider the goal:

- set_goal([‘‘n = 1‘‘, ‘‘m = 2‘‘], ‘‘n < m‘‘);
> val it = New goal stack.
n < m
____________________________

m = 2
n = 1

: proofs

x The tactic SUBST1_TAC:

A ?- P

=============== SUBST1_TAC (B |- t = u)

A ?- P[u/t]

- SUBST1_TAC;
> val it = fn : thm -> tactic

Example - concluded

x Substitute into the conclusion:

- expand (EVERY_ASSUM SUBST1_TAC);
OK..
1 subgoal:
> val it =
1 < 2
____________________________

m = 2
n = 1

: proofs

Using the First Useful Assumption

x The tactical

FIRST_ASSUM : (thm -> tactic) -> tactic

takes an ML function f of type thm -> tactic

and a goal (A,g), and

u maps ASSUME down the list of assumptions A,

u maps f down the resulting list of theorems,

u applies the first tactic in the resulting list of tactics

that succeeds when applied to the goal.

x For example, consider the goal:

- set_goal ([], ‘‘!R Q. (R /\ Q) ==> (Q /\ R)‘‘);
> val it = New goal stack.
!R Q. R /\ Q ==> Q /\ R
____________________________

: proofs

Example Continued

x After stripping:

- expand (REPEAT STRIP_TAC);
OK.. 2 subgoals:
> val it =
R
____________________________

R
Q

Q
____________________________

R
Q : proofs

x The conclusion is an assumption, so:

- expand (FIRST_ASSUM ACCEPT_TAC);
OK..
Goal proved.
. |- Q

Remaining subgoals:
> val it =
R
____________________________

R
Q : proofs

HOL/FM: S-10 c© T F Melham 1993 Tacticals: 21

Using the Entire Assumption List

x The tacticals FIRST_ASSUM and EVERY_ASSUM are
special cases of

ASSUM_LIST : (thm list -> tactic) -> tactic

which takes an ML function f of type

thm list -> tactic

and a goal (A,g), and

u maps ASSUME down the list of assumptions A,

u applies f to the resulting list of theorems,

u applies the resulting tactic to the goal.

x For example, EVERY_ASSUM is defined by:

- val EVERY_ASSUM:thm_tactic->tactic =
ASSUM_LIST o MAP_EVERY;

> val EVERY_ASSUM = fn : thm_tactic -> tactic

Summary — List Tacticals

x For lists of tactics:
u EVERY : tactic list -> tactic

u FIRST : tactic list -> tactic

x Mapping tacticals:

u MAP_EVERY : (’a -> tactic) -> ’a list -> tactic

u MAP_FIRST : (’a -> tactic) -> ’a list -> tactic

x Using assumptions:

u EVERY_ASSUM : thm_tactic -> tactic

u FIRST_ASSUM : thm_tactic -> tactic

u ASSUM_LIST : (thm list -> tactic) -> tactic

‘Resolution’ Tactics

x In HOL, ‘resolution’ involves deducing from a
list of facts

[|- P1, |- P2, ..., |- Pn]

and an implication

|- Q1 ==> (Q2 ==> ...==> (Qm ==> Q))

some instance of |- Q.

x This is done by matching antecedents Qj with
facts Pi and using modus ponens.

Matching Modus Ponens

x Suppose we have two theorems

thm1 = |- !x:’a. !y:’a. !z:’a. R[x, y, z] ==> Q[x, y, z]

thm2 = |- R[a:num, b:num, c:num]

and wish to deduce |- Q[a, b, c]

x This could be done as follows:

- val ith = INST_TYPE [(‘‘:num‘‘, ‘‘:’a‘‘)] thm1;
> val ith = |- !x y z. R[x,y,x] ==> Q[x,y,z]

- val sth = SPECL [‘‘a:num‘‘, ‘‘b:num‘‘, ‘‘c:num‘‘] ith;
> val sth = |- R[a,b,c] ==> Q[a,b,c]

- val thm3 = MP sth thm2;
> val it = |- Q[a,b,c]

x But MATCH_MP does this automatically:

- val thm3 = MATCH_MP thm1 thm2;
|- Q[a,b,c]

Resolution

x The main resolution tactic is

IMP_RES_TAC : thm -> tactic

This takes an implication

|- R1 ==> (R2 ==> ...==> (Rn ==> Q))

and does the following:

u match the antecedents R1, . . . , Rn to the

assumptions of the goal

u derive conclusions by modus ponens and add

the conclusions to the assumption list.

x The tactic:

RES_TAC : tactic

does resolution among the assumptions of a goal
by calling IMP_RES_TAC for all the implications
in the assumption list.

Example of Resolution

x Example:

- set_goal ([‘‘a < b‘‘, ‘‘b < c‘‘], ‘‘~(c < a)‘‘);
> val it = New goal stack.
~(c < a)
____________________________

b < c
a < b : proofs

- val [_,thm] =
RES_CANON arithmeticTheory.LESS_TRANS;

> val thm = |- !p n. n<p ==> (!m. m<n ==> m<p):thm

- expand (IMP_RES_TAC thm);
OK.. 1 subgoal:
> val it =
~(c < a)
____________________________

b < c
a < b
!m. m < a ==> m < b
a < c : proofs

x If the matching can be done several ways, all
possible results are added to the assumptions.

Example Continued

x IMP_RES_TAC can solve goals:

- set_goal ([‘‘a < b‘‘, ‘‘b < c‘‘], ‘‘(a < c)‘‘);
> val it = New goal stack.
a < c
____________________________

b < c
a < b : proofs

- arithmeticTheory.LESS_TRANS;
> val it = |- !m n p. m < n /\ n < p ==> m < p : thm

- expand (IMP_RES_TAC it);
OK..
> val it =
Original goal proved.
.. |- a < c : proofs

Solving Goals with Resolution

x A goal is solved if resolution generates:

u the theorem Γ |- F, or

u a theorem α-equivalent to the goal’s conclusion

x Example proof:

- set_goal
([], ‘‘((!x:’a. R x ==> Q x) /\ R a) ==> Q a‘‘);

> val it = New goal stack.
(!x. R x ==> Q x) /\ R a ==> Q a
____________________________

: proofs

- expand (REPEAT STRIP_TAC);
OK.. 1 subgoal:
> val it =
Q a
____________________________

!x. R x ==> Q x
R a : proofs

- expand RES_TAC;
OK.. Goal proved.
.. |- Q a
> val it =
Original goal proved.
|- (!x. R x ==> Q x) /\ R a ==> Q a : proofs

Summary — Resolution

x The matching modus ponens rule:

u MATCH_MP : thm -> thm -> thm

x The resolution tactics:
u IMP_RES_TAC : thm -> tactic

u RES_TAC : tactic

x Some related tools:
u canonical form: RES_CANON

u one-step resolution: IMP_RES_THEN

u one-step assumption resolution: RES_THEN

See the manual for details.

An Example Tactic Proof

x The goal to prove is:

- set_goal([], ‘‘!m n. (n<=m) ==> ?p. m=n+p‘‘);
> val it = New goal stack.
!m n. n <= m ==> (?p. m = n + p)
____________________________

: proofs

x Expand with the definition of <= and strip:

- expand(PURE_ONCE_REWRITE_TAC [arithmeticTheory.LESS_OR_EQ]);
OK.. 1 subgoal:
> val it =
!m n. n < m \/ (n = m) ==> (?p. m = n + p)
____________________________

: proofs

- expand(REPEAT STRIP_TAC);
OK..
2 subgoals:
> val it =
?p. m = n + p
____________________________

n = m

?p. m = n + p
____________________________

n < m
: proofs

Example Continued

x We can use the fact LESS_ADD_1:

- arithmeticTheory.LESS_ADD_1;
> val it = |- !m n. n<m ==> (?p. m = n + p + 1) : thm

- expand(IMP_RES_TAC arithmeticTheory.LESS_ADD_1);
OK..
1 subgoal:
> val it =
?p. m = n + p
____________________________

n < m
m = n + p + 1

: proofs

x We now substitute for m:

- expand (FIRST_ASSUM SUBST1_TAC);
OK..
1 subgoal:
> val it =
?p’. n + p + 1 = n + p’
____________________________

n < m
m = n + p + 1

: proofs

Example Continued

x The proof of this subgoal is now easy:

- expand (EXISTS_TAC (‘‘p + 1‘‘));
OK..
1 subgoal:
> val it =
n + p + 1 = n + p + 1
____________________________

n < m
m = n + p + 1

: proofs

- expand REFL_TAC;
OK..
Goal proved.
.. |- n + p + 1 = n + p + 1
Goal proved.
.. |- ?p’. n + p + 1 = n + p’
Goal proved.
.. |- ?p. m = n + p
Goal proved.
. |- ?p. m = n + p

Remaining subgoals:
> val it =
?p. m = n + p
____________________________

n = m
: proofs

Example Continued

x The correct value is 0:

- expand (EXISTS_TAC (‘‘0‘‘));
OK..
1 subgoal:
> val it =
m = n + 0
____________________________

n = m : proofs

x An important theorem helps now:

- val ADD_THMS = arithmeticTheory.ADD_CLAUSES;
> val ADD_THMS =
|- (0 + m = m) /\

(m + 0 = m) /\
(SUC m + n = SUC (m + n)) /\
(m + SUC n = SUC (m + n)) : thm

- expand (ASM_REWRITE_TAC [ADD_THMS]);
OK..
Goal proved. . |- m = n + 0
Goal proved. . |- ?p. m = n + p
Goal proved. |- !m n. n < m \/ (n = m) ==>

(?p. m = n + p)
> val it =
Original goal proved.
|- !m n. n <= m ==> (?p. m = n + p) : proofs

Example Concluded

x The entire proof is:

- val thm =
TAC_PROOF
(([], ‘‘!m n. (n<=m) ==> ?p. m=n+p‘‘),
PURE_ONCE_REWRITE_TAC [arithmeticTheory.LESS_OR_EQ] THEN
REPEAT STRIP_TAC THENL
[IMP_RES_TAC arithmeticTheory.LESS_ADD_1 THEN
FIRST_ASSUM SUBST1_TAC THEN
EXISTS_TAC ‘‘p + 1‘‘ THEN
REFL_TAC,
EXISTS_TAC ‘‘0‘‘ THEN
ASM_REWRITE_TAC [arithmeticTheory.ADD_CLAUSES]]);

> val thm = |- !m n. n <= m ==> (?p. m = n + p) : thm

